

赤緯軸駆動プログラム ソースコード

```
//*****//
:
:-----
: USB-IOでステッピングモーター駆動
:-----
:
: ポート 0 の上位4ビットで2相励磁パルス発生
: ポート 0 の下位2ビットで回転方向検出(active high)
:     ビット0= 0 で左回転
:     ビット1= 0 で右回転
:     ビット2= 0 でDecデータラッチ
:
//*****//

#include "hspusbio.as"
#packopt name "ステッピングモーター駆動 完成版" ;実行ファイルネーム指定

screen 0,300,300,0,275,160           ;画面サイズ
x=150:y=200:z=200                   ;背景色指定(RGB)

color x,y,z:boxf
gsel 0,1                             ;0(0は最初に表示されるウィンドウ)ウィンドウを通常表示(2で常に手前に表示)

title "ステッピングモーター駆動 ラッチ版" ;タイトルバー表示指定
a=6                                   ;回転周期の初期値設定
pos 130,89
input a,22,20                         ;回転周期aの入力と入力ボックスの大きさ指定

pos 90,93: color 0,0,0;font "MS ゴシック",13
mes "周期"
pos 155,93
mes "x60秒"

pos 115,150
button "起動",*start                 ;起動ボタンの表示と、クリックでラベルにジャンプ

pos 115,200
button "停止",*mkstop                ;停止ボタンの表示と、クリックでラベルにジャンプ

uio_find
if stat=2:goto *search                ;USBIOが接続されていない場合、ラベルにジャンプ

*start
wait a                                ;待機
uio_out 0,251                          ;11111011 ポート0を入力に設定(Decデータラッチ。0xF?でモーター停止)
uio_inp g,0                             ;ポート0のデータ取り込み
if g=250 : goto *south                 ;11111010 0ビットが"0"(1ビットが"1")で左回転(*south)にジャンプ
if g=249 : goto *north                 ;11111001 1ビットが"0"(0ビットが"1")で右回転(*north)にジャンプ
uio_out 0,255                          ;11111111 データラッチ解除(ビット2を 1)
goto *start                             ;繰り返し

// 左回転
*south
uio_out 0,207                          ;11001111 出力 (X,Y出力) 下位4ビットはデータを取り込み用 0x?F
wait a
uio_out 0,159                          ;10011111 出力 (Y,-X出力)
wait a
uio_out 0,63                           ;00111111 出力 (-X,-Y出力)
wait a
uio_out 0,111                          ;01101111 出力 (-Y,X出力)
goto *start                             ;ラベルにジャンプ
```

```

// 右回転
*north
  uio_out 0,63          ; 00111111 出力 (-Y,-X出力) 下位4ビットはデータを取り込み用 0x?F
  wait a
  uio_out 0,159        ; 10011111 出力 (-X,Y出力)
  wait a
  uio_out 0,207        ; 11001111 出力 (Y,X出力)
  wait a
  uio_out 0,111        ; 01101111 出力 (-Y,X出力)
  goto *start          ; ラベルにジャンプ

// USB接続確認
*search
  repeat
  pos 35,250
  color 250,0,0
  font "MS ゴシック",15,1
  mes "USB-I/Oを接続して下さい" ; 確認メッセージの表示
  wait 60
  color x,y,z:boxf 30,230,280,290 ; 背景色でメッセージを消す
  wait 50
  uio_find              ; USBIO接続チェック
  if stat=0:goto *start ; 接続されるとラベルにジャンプ
  loop
stop

// 停止
*mkstop
  uio_out 0,255        ; 11111111 ポート0に 0xFF を出力(モーター停止)
  color x,y,z:boxf 30,230,280,290 ; 表示消去(背景色で塗りつぶす)
stop

//終了
onexit goto *mkend    ; 終了ボタンXが押されるとラベルにジャンプ
*mkend
end

```